
Overview
Summary

So what exactly is Axe Parser?  Axe is a revolutionary new programing language for the 
TI-83/84  series  calculators.   It  is  designed  with  game  creation  in  mind,  but  the 
applications are limitless.  It offers an alternative to the restrictions of BASIC but without  
the  complexity  of  pure  Assembly.   An  Axe  program  can  compile  into  a  no-stub 
executable  or  to  your  favorite  shell  and  therefore  does  not  require  any  external 
application to run your program.

Axe vs. Other Languages
BASIC xLIB/Celtic BBC Basic Assembly Axe

Language 
Difficulty Easy Easy Medium Hard Medium

Speed Slow Medium Fast Very Fast Very Fast
Editable
On-calc? Yes Yes Special Editor 

Required Somewhat Yes

Execution Interpreted Interpreted Interpreted Compiled Compiled
Sprite 
Support No Yes Yes Yes Yes

External 
Variables
Required
To Run

Pictures, Vars,
Lists, Strings,
etc.

Same as 
BASIC plus
16kb App

49kb App None None

Shell 
Compatibility Yes Some No Yes Yes

Specialty Math Games Variety Everything Games
Source Code
Visible Always Always Always Optional Optional

There  is  no  one  programming  language  for  everything,  each  has  advantages  and 
disadvantages.  It is up to you to decide what best fits your needs.  If your top priorities 
are speed, quick development, graphics, and calculator portability, then Axe Parser is 
definitely the way to go.

Freedoms & Limits
Axe Parser allows you to do what most calculator programming languages do not.  You 
are able to draw sprites, access external variables (like appvars), multiplayer linking, 
grayscale  support,  interrupts,  sound,  multi-key press  detection,  contrast  adjustment, 
and much much more.  All of this usually at the same speed as regular assembly.  You 
even have the option of inserting your own asm code directly into the program.

Unfortunately, with every new freedom, there is a price to pay.  Like assembly, but unlike 
the other languages, a bad crash in Axe will usually cause a RAM clear and you will 
lose most of your data.  If the crash is really really bad, its possible but unlikely, that the  
archive could get corrupted as well.  For this reason, it is recommended that you begin 
programming on an emulator until you get used to the language enough to where you 
don't  make  those  kinds  of  mistakes.   I  recommend  wabbitemu  which  can  be 
downloaded at revsoft.org.  Another option is to remember to archive your source code 
and backup all your programs before you start experimenting with Axe Parser.

Another  thing  to  keep  in  mind  is  that  the  programs are  not  as  optimized  as  pure 
assembly.   They will  be on average one and a half  to  two times larger  than if  the  
program were actually optimized by an experienced asm programmer.  However, there 
are many optimizations that can be done using Axe itself, and that will be discussed in a 
later section.

Axe is the ultimate tool for a good execution (of programs)

Using Axe Parser
Writing & Editing Programs

To start making your programs, you follow the exact same procedure as if you were 
starting a BASIC program.  You go to [PRGM], New, and then type in the name.  This is  
your source code so make sure you pick a name that's not the same as what you plan 
to name your final program.  For instance if you plan to make “MARIO” you might want 
to name your source “MARIOSRC”.

The next thing you absolutely need is an Axe Header.  You must start the first line with a  
period, which is a comment in Axe, followed by the name you want for the compiled 
program.   If  you  want  a  program description,  you  can  type  a  space  and  then  the 
description on that same line, but this is optional.  In the above example your program 
might look like this:

PROGRAM:MARIOSRC
:.MARIO A fun platformer
:

That's basically all you need to get started.  This program should show up on the list of  
programs in the Axe Compile menu.  You can compile from both RAM and Flash.

There is also another kind of Axe header which starts with two periods instead of one. 
These are library files that have the special property that they do not show up on the 
compiling  list  so  they  cannot  be  compiled  independently;  only  as  part  of  another 
program.  However, you can still import regular Axe source code as a library.

Compiling Programs
To compile a program, go ahead and open up the Axe Application.  In “Options” you can  
select what type of shell you want to compile for.  If you don't want a shell, you can run  
your programs using the “asm(“ command in the catalog.  There is also an option to 
compile to applications.  This uses very hackish code and may not work on all models of  
calculator so be cautious when using this feature as this is the one that can cause 
archive corruptions if the it fails to write the application correctly.

Next, select “Compile” to see a list of all Axe source code files on your calculator.   You  
can use the arrow keys or any letter key to select your program, and then press [Enter]  
to compile.  Compiling usually takes about one second for every 2000 bytes of source 
on an 84+ so don't be surprised if its almost instant.  If you get an error, check the error  
codes section to diagnose the problem.

You can also press [zoom] instead of [Enter] to compile quickly, but the executables will  
be slightly larger and slightly slower.  This feature is mainly to test changes in large 
programs if you don't want to wait several extra seconds to see if the change worked.

After an error, Axe will scroll to the error just like in BASIC (unless you press [clear]) if  
the source is unarchived and the error was a problem with the source code.  Once 
compiling is complete, you should see the executable in your programs list.

Since running untested programs can often be unsafe and lead to loss of source code 
Axe provides a quick way to backup your source code.  If  you press [Alpha] in the 
compile menu, a copy of the selected program is created in archive so that just in case 
your source gets deleted or corrupted, you can recover it from the last backup.  Once  
backed up, you will see a copy of the program appear in the compile list but with a hash 
next  to  it  indicating  that  its  a  backup file.   Selecting  it  will  restore  the backup and 
pressing [Del] will delete the backup file.

You also have the option of making the backup process automatic.  If you select “Auto 
Backup” from the safety section of the options menu, then a new backup will be created 
every time you compile the source.

   Error Messages
BAD CONSTANT A non-constant was found where a constant was expected.
CANNOT USE HERE Valid syntax, but not useable in the current environment.
DECLARE FIRST Constants, unlike data, must be declared before they're used.
DUPLICATE SYMBOL The label or static pointer already exists.
ELSE ONLY FOR IF Else can only be used after an if or elseif statement.
INVALID AXIOM You are using a corrupted or missing Axiom.
INVALID HEX The hexadecimal number is invalid or the wrong size.
INVALID NUMBER The number is too large or small to fit in 2 bytes.
INVALID TOKEN The token used is unsupported here.
INVLAID FILE USE You have used a file type incorrectly.
LABEL MISSING You called a label or subroutine that does not exist.
LOW BATTERY You cannot write to archive if the battery is low.
MISSING AN END A loop or if statement was not closed by the end of the program.
MISSING PROGRAM The library program you tried to include is not an Axe file.
MUST END COMMENT A multi-line comment was not capped by the end of the source.
NAME LENGTH Name is too long for a label or static pointer
NO NESTED LIBS You cannot have nested library files at the moment.
OS VAR MISSING The object that is to be absorbed does not exist.
OUT OF MEMORY The ram is full and there is no more room to write the program.
PARENTHESIS An ending parenthesis is required for this command.
SAME OUTPUT NAME You named the output file the same as your source.
TOO MANY AXIOMS Only 5 Axioms are allowed total per program.
TOO MANY BLOCKS There are too many nested loops or if statements.
TOO MANY ENDS Found an “End” that doesn't end anything.
TOO MANY SYMBOLS There are too many static pointers or labels.
TOO MUCH NESTING There are too many parenthesis in a single expression.
UNDEFINED The static pointer you are using was not declared.
UNDOCUMENTED The Axiom used an undocumented instruction.
UNKNOWN ERROR Something is wrong with the parser.  Report bug immediately!
WRONG # OF ARGS You have the wrong number of arguments.

Programming
Numbers & Basic Math

Alright, now to the fun stuff!  Okay, this is one of the most important differences between 
Axe and BASIC.  Numbers in Axe are all 16-bit integers meaning that there's no such 
thing as fractions and decimals.  What the 16-bit part of it means is that a number can 
only hold a value between 0 and 65,535.  This is called the unsigned number system 
meaning that there is no sign:  all the numbers are positive.

Now wait a minute you say, what if I want to use negative numbers?  Well in that case, 
you want to use signed representation.  The way that works is that we cut our range in 
half and say that all the numbers on one side are positive and numbers on the other 
side are negative.  So numbers from 0 to 32767 we still say are positive but now the 
numbers from 32768 to 65535 we say are actually -32768 to -1.  So our new range is 
-32,768 to 32,767

Remember, both representations are really the exact same number.  It's just a different  
way of representing it.  So -1 really is the same number as 65535.  And I'm not just  
making this up, the mathematics works this way.  If you add 65535 to a number you get 
exactly the same result as if you subtracted 1!  How the heck does that happen?  That 
brings me to my next point which is modular arithmetic.

Remember, we can only hold values between 0 and 65535 right?  Well what happens if  
we keep adding and overflow the maximum value?  Let's count by constantly adding 1:

0,1,2,3,...,65533,65534,65535,0,1,2,3...

Do you see what happened?  Once you pass the maximum, you loop around back to 0 
again.  Signed representation does the same thing, best illustrated by this xkcd comic:

Now lets  get  into  some Axe code to  apply our  new-found knowledge.   This  will  all 
appear  to  be very similar  to  BASIC, but  remember,  the number system works  in  a 
completely different  way.   Lets make a program that  finds the sum of  the first  100 
numbers:

PROGRAM:A
:.SUM100
:0→S
:For(N,1,100)
:S+N→S
:End
:Disp S Dec▶

You have to add Dec to▶  
display numbers because 
there are multiple ways 
to display data in Axe. 
Numbers display using 
unsigned representation.

Next up is multiplication and division.  Multiplication is nice because mathematically, its  
just repeated addition, so it works the same for signed and unsigned numbers.  Division 
on the other hand can give different results depending on if you want to do it signed or 
unsigned.  Signed division, or any signed math in general, you can perform by typing 
the operation twice.  For instance:

-10/5 = 13105
-10//5 = -2

Parser sees: 65526/5
Parser sees: -10/5

Although the math is basically what you would expect, what you might not expect is that 
Axe always does the order of operations from left to right always.  So watch carefully 
how the parser will read this:

4-3*8+2/3
1*8+2/3
8+2/3
10/3
3

Subtraction first
Multiplication
Addition
Division
Always rounds down.

Of course,  you can use parenthesis whenever  you need them.  The reason I  don't 
automatically add the parenthesis for you to do the usual order of operations is because 
I want you to be aware of how many parenthesis you use.  The less you need, the more 
optimized the code is and the faster and smaller it will be.  You'll see more about this in 
the optimization section.

Another thing I should mention is that the caret key “^” no longer means “power”.  It now 
is the modulus operator.  It means the remainder left over after a division.  You can learn 
more about this command and the other advanced math commands in the commands 
list section.

Pointers
You must  understand pointers!   If  you've never  used a programming language with  
pointers before, pay VERY close attention.  Pointers are such powerful and useful tools 
that you'll be using them in virtually all of your programs.  We'll start slow.

What is RAM?  You probably know its just a bunch of memory, 65536 bytes of it to be 
exact on the TI-83/84.  But how do you access all of that RAM?  How do you read and 
write to it?  We just follow the simplest procedure; give every byte in RAM its own 
address.  The 0th byte has the address 0, the 1st byte has the address 1, all the way up 
to the last byte with the address 65535.  Get used to starting your counts at zero by the 
way.

So lets make some data.  We'll start with the classic string:

PROGRAM:A
:.HELLO
:"Hello World"→Str1
:Disp Str1

Its just like BASIC right?  Wrong!  It just  appears to behave like BASIC.  Let's look at 
what's actually going on.  First of all, Str1 is a pointer which is a number, not an actual 
string.   The  sentence  “Hello  World”  gets  stored  at  an  address  somewhere  in  the 
executable  and  the  pointer  Str1  is  just  a  number  that  tells  us  where  that  string  is 
located.    So yes! Display really does just take a number as an argument!  Then it takes 
that pointer, finds the data it points to, and that's how it knows what string to display.

But since a pointer is just a number, we can do math with it!  What do you think would  
be the effect of doing this modification?

PROGRAM:A
:.HELLO
:"Hello World"→Str1
:Disp Str1+6

This just outputs “World”.  That's because the string is stored in order in consecutive 
addresses.  Lets pretend the “H” was located at address 100.  That means “e” is at 101,  
the first “l” is at 102, etc.  So by the time we get to the address 106, we're at “W”.  That's  
why the display routine skips the hello part.

In Axe, Str1 and the other calculator variables are called  static pointers. What that 
means is once you declare them, their value cannot change for the rest of the program.  
The allowed static pointer variables are Str,  Pic,  and GDB.  You can use these for 
whatever you want since they're all just numbers anyway, but its a naming convention to 
use Str for strings, Pic for sprites, and GDB for data.  None of these are the BASIC 
variables by the way, they're entirely new variables that just share the same name for 
convenience.

Another thing is that Axe allows you to name everything with up to 5 numbers/letters 
instead of just a single number.  The following are all unique and valid names:

Str1
Pic0
GDB4AXE

Str01
Pic9ZZZZ
GDB45

Str666
Pic8C
GDB3X

There are other ways to enter data, one of those is by using hexadecimal.  You don't 
really have to worry about how it works, but its just a more compact way to write a 
number since there's 16 characters for each digit instead of 10.  This is convenient for 
sprites since you can use a tool to convert them; one is included in the package.

So now lets draw a happy face sprite.

PROGRAM:A
:.HAPPY
:[3C7EDBFFBDDB663C]→Pic1
:ClrDraw
:Pt-On(44,28,Pic1)
:DispGraph

Pt-On Draws sprites by 
taking the position on the 
screen and a pointer to 
the sprite as arguments.
Then we have to update the 
screen to see it.

Data & Arrays
So how do we actually manipulate data?  First, let's take care of reading.  We have to 
tell the calculator what byte to read, so we need to give it the address.  Let's look at how 
we can read a list of numbers:

PROGRAM:A
:.LISTREAD
:Data(5,2,6,11,4)→GDB1
:For(A,0,4)
:Disp {A+GDB1} Dec▶ ,i
:End

ΔList is data in the form 
of a list of numbers.

The imaginary 'i' is a 
newline character.

We use the curly brackets to indicate that we want the byte that's at the address of the  
pointer, not the the value of the pointer itself.  This is what the pointer is said to “point to”  
in a process called referencing.

We can similarly store values into these addresses:

PROGRAM:A
:.LISTWRIT
:Data(0,0,0,0,0)→GDB1
:For(A,0,4)
:A*2→{A+GDB1}
:Disp {A+GDB1} Dec▶ ,i
:End

Remember, bytes stored in 
memory can only hold 
values between 0 and 255 
unsigned or -128 to 127 
signed.  If you want to 
have the full range, see 
the {}r command.

Even though you wrote over data in the program, that data isn't going to be saved when 
you exit.  That's because when you run a program, it actually executes a copy of the  
program  and  when  its  done  executing  there's  no  write-back meaning  that  the 
temporary copy just disappears and doesn't replace the original.

Its really pointless to have to store this list, called an array in the program itself since its 
data isn't actually used and it gets written over anyway.  Instead, the calculator actually  
has some special addresses called free ram that you're allowed to use for whatever you 
want.  The addresses to the largest of these locations are in the constants L1-L6.  You'll 
have to see the command list for more details about which ones are best to use.  I'll use 
L1 and L2 in my examples.

Lets say you need to control 20 sprites on the screen and each sprite needs an X and Y 
coordinate that has to be stored and updated.  Here is an example of what you can do:

PROGRAM:A
:.DRAW20
:[3C7EDBFFBDDB663C]→Pic1
:
:.Initialize XY
:For(A,0,19)
:rand^88→{A+L1}
:rand^54→{A+L2}
:End
:
:.Display Spites
:For(A,0,19)
:Pt-On({A+L1},{A+L2},Pic1)
:End
:DispGraph

The easiest way to get 
random numbers is to take 
the modulus with X to get 
a random number between 0 
and X-1.

Also, when you're using 
free ram, it DOES NOT use 
any program ram like in 
BASIC.  A list or matrix 
in free ram takes 0 
memory whereas the same 
data in BASIC can be 
pretty huge.

Matrices, or more generally multidimensional arrays are a little trickier.  Imagine a 2 
dimensional array that we want to store in L1.  We have to fit this into a one dimensional 
array so the best way to do it is by combining it from left to right and top to bottom.

1  2  3
A  B  C
X  Y  Z
L  O  L

  If we know the row “R” and column “C” both starting at 0, we can find the data in that 
position by using this simple math trick:

So in general, you can use this formula, which is already in order of operations:
<Row#>*<Total Columns> + <Column#> + <Start of Data>

Read & Write to External Variables
There are times when you need to save information that will continue to be there even 
after you quit the program like high scores for instance.  You may also want to read off 
or  write to external  data from the calculator  such as pictures or strings.  There are 
currently 2 methods for accomplishing these tasks.  You can either read and write data  
from RAM or you can read data directly from Archive.

     Reading Variables From RAM

Since RAM is random access memory, you can read, write, and modify variables freely.  
The main restriction with RAM however, is that it's relatively small, gets erased during 
ram clears, and can easily be corrupted by bad coding.  However, once a variable is 
generated in RAM, you always have the option of archiving and unarchived it from your 
program.

In order to do anything with variable manipulation, you need to know the name of the 
variable you intend to use.  Since names can be long, the entire name cannot fit into a  
single variable,  so instead,  routines involving external  variables will  take pointers to 
Name Strings.  The following chart lists some examples of name strings:

"prgmABC"
"appvABC"
"grpABC"
"LABC"
"L1"
"varA"
"Str1"
"GDB1"
"Y1"
"[A]"
"Ans"

The Program ABC
The Appvar ABC
The Group ABC (Archive Only)
The List ABC
The List L1
The Real number A
The String Str1
The Graph database GDB1
The Equation Y1
The Matrix [A]
The Ans variable

Some types of name strings need prefixes and those are underlined in the chart above. 
They are NOT the lowercase letters, they are specific tokens, most of which can be 
found using [2nd] and then [7], [8], or [9] which are the u, v, and w recursive tokens 
respectively.

Unfortunately, unlike free ram, external variables are not in the same location every time 
you run the program and thus can not just be located with a static pointer.  Instead, you 
have to use the Axe command GetCalc to retrieve a pointer to the variable.  Once you 
have that pointer, you can store it to a variable and use it for the rest of the program.

For example, lets say we wanted to Retrieve the appvar “MyScore” from ram to see if  
our score S has beat the high score or not.

PROGRAM:A
:.HISCORE
:"appvMyScore"→Str1
:GetCalc(Str1)→P
:If P
: If S>{P}r

:  S→{P}r

: End
:End

If statements don't need 
the “Then” anymore.

We use the r modifier to 
store the full 16-bit 
number instead of just 
the first 8-bits.

Notice we also have to make sure the appvar exists before we can use it.  That's what 
the “If P” is for.  It just makes sure P is non-zero.  If it were zero, it means that either the 
appvar is not in ram or it doesn't exist.  Speaking of existence, how do we create the 
appvar in the first place?  Let's modify the example to create the appvar if it doesn't  
exist and set the initial high score to zero in that case.

PROGRAM:A
:.HISCORE
:"appvMyScore"→Str1
:!If GetCalc(Str1)→P
: GetCalc(Str1,2)→P
: Return!If P
: 0→{P}r

:End
:If S>{P}r

: S→{P}r

:End

The second argument is the 
size.  Its only 2 in the 
example because we only 
have a single 2 byte number 
to store in the appvar.

If the pointer is still 
zero after trying to create 
it, there is not enough RAM 
left on the calculator. So 
it will just quit.

The size argument here makes sense because appvars can be different sizes.  But 
what about things like real numbers?  Even though they have a fixed size, you still need 
to specify the correct size because the program will use that number to determine if  
there is enough RAM to create the variable.

     Reading Variables From Archive

Reading from Archive is slightly different than reading from RAM.  The main difference 
is  that  instead  of  using  pointers,  you  use  Files.   You  can  have  up  to  10  files 
simultaneously opened at once.  Each file is represented by one of the “Y-vars”: Y 0 – Y9 

Luckily,  even  though  files  are  completely  different  than  pointers  due  to  technical 
reasons, the Axe syntax allows you to use them as if they were regular pointers in two 
of the most important commands:  the curly brackets and the copy command.

Here is the same high score example as before, but this time, the score will be read 
from archive and then displayed:

PROGRAM:A
:.HISCORE
:"appvMyScore"→Str1
:GetCalc(Str1,Y1)→A
:Return!If A
:Disp {Y1}r

GetCalc does NOT return a 
pointer when opening a file. 
Its simply non-zero if it 
exists in archive and 0 
otherwise.

The Maximum size file you can read from is 48kb but I don't think any variable gets that 
big  anyway except  for  groups maybe.   Each type  of  external  variable  has  its  own 
structure.  Sometimes its simple like appvars, programs, strings, and pictures, where its 
nothing but pure data.  Sometimes the formatting is weird like for real numbers, lists and 
matrices.  I don't even know how groups are formatted to be honest, you may want to 
check out the TI-Developer's Guide for more information on variable formatting.

Another thing I should mention is that when reading files from either RAM or Archive, 
the size of the variable you are opening can be found 2 addresses before the start of  
the data.  For instance, if you were unsure about the size of the appvar in the above  
example, you can get the size with the following code:

:{Y1-2}r→S Saves size of the Y1 file to S

An Example Program
Let's make one of the first video games ever created: Pong!  Fist our simple header:

PROGRAM:PONGSRC
:.PONG MY FIRST AXE GAME

Simple Description

Next step is to define all our data.  There will be 2 strings and 2 sprites:

:"PONG"→Str1
:"SCORE:"→Str2
:[000000000000FFFF]→Pic1
:[0000182C3C180000]→Pic2

Our Title
For the score
The paddle
The ball

Now a very primitive title screen.  It will just say “PONG” and pause for a second.

:DiagnosticOff
:ClrHome
:Output(6,3,Str1)
:Pause 1000

You should almost always 
include the DiagnosticOff 
command in start of your 
programs because it makes 
them much cleaner.

Now to initialize our variables.  The ball's X position will be inflated for more precision.

:0→S-1→D
:44→Z*256→X
:10→Y
:sub(HT)

S=Score, D=YSpeed
Z=Paddle_X, X=Ball_X*256
Y=Ball_Y, V=XSpeed
HT will be a subroutine 
that is called when the 
ball hits the paddle and 
it sets the new speed.

Setup the main loop of the program.  Also take care of all key presses.  

:Repeat getKey(15)
:If getKey(2) and (Z≠0)
:Z-2→Z
:End
:If getKey(3) and (Z≠88)
:Z+2→Z
:End

By testing individual 
keys rather than all of 
them at once, it makes 
the program a lot faster 
and more flexible.

Page 17



                             Axe Parser  1.1.2      

Now, we update the new position of the ball.

:X+V→X
:Y+D→Y

Position plus speed
equals new position

Here is the collision detection.  Bounce if you hit a wall or the paddle and change the 
speed.  Quit if the ball goes off of the screen.

:If Y>70
:Goto D
:End
:If Y=0
:sub(HT)
:End
:If Y=54 and
 (abs(X/256-Z)<8)
:sub(HT)
:S+1→S
:End
:If X/256=0 or (X/256=88)
:-V→V+X→X
:End

Quit when reaching edge. 
D is the ending label.

Bounce off the roof and 
change direction.

If the paddle and ball 
are close enough, bounce 
and increase the score.

Bounce off of a wall and 
update the position 
again.

Time to draw the sprites and update the screen:

:ClrDraw
:Pt-On(Z,54,Pic1)
:Pt-On(X/256,Y,Pic2)
:DispGraph

The buffer only updates 
to the screen when you 
call DispGraph

We're done with our loop, so lets add the quit code:

:End
:Lbl D
:ClrHome
:Disp Str2,S Dec,▶ i
:Return

End the loop.

What used to be “Stop” in 
BASIC is now “Return”

Finally, we'll add our “Hit” subroutine to bounce the ball:

:Lbl HT
:rand^512-256→V
:-D→D

Subroutines should end 
with “Return” but the end 
of the code automatically 
returns anyway, so no 
reason to add it twice.

That's it!  So just run it through the parser and you should see a new “PONG” program 
show up on your program list.  If you didn't use a shell, you can run it by using the 
command asm().  The result is a fun little game that should keep you occupied for many 
years to come!  (Hopefully this estimate is an exaggeration)

Anyway,  the  outline  here  for  program  flow  is  a  pretty  standard  practice  for  game 
programming.  You have in order: the Header, data declarations, title screen, variable 
setting, the main loop, quitting code, and then all  your subroutines.  There are a lot  
more example programs included with the parser, you can check out their source for a  
better look at exactly how its done.

Good luck with your projects and happy programming!

Optimization Tricks
THE FIRST LAW OF OPTIMIZATION: 99% OF THE TIME, IF THE COMPILED CODE 
IS SMALLER, ITS ALSO FASTER.  Basically, you'll have to play around with different 
commands.  Start a new program and type out the command you are trying to optimize 
and compile it.  Check the size of the  compiled program and then try another way of 
doing it until you find the smallest size.

If you're not going to use a subroutine more than once, just put it inline.  Even if it might 
make the program more organized, it is adding at least 4 bytes to the file size.

Conversely, when you do find yourself repeating the same actions twice or more, its  
probably best to put it in a subroutine.  You can always test the program both ways and 
see if it makes it any smaller.

If you have constants in your expression, try putting them at the end if at all possible.

Unoptimized
:2*A
:1+A

Optimized
:A*2
:A+1

Try to make expressions use the least amount of parenthesis as possible.  Use left to 
right order of operations  to your advantage.

Unoptimized
:A*(B+(C*D))

Optimized
:C*D+B*A

When initializing variables, initialize variables with the same value together. Also, if the 
variables are at most 2 apart, then is also yields an optimization to do this.

Unoptimized
:0→A:0→B:0→C
:10→A:11→B:13→C

Optimized
:0→A→B→C
:10→A+1→B+2→C

If the last line of your entire code is a return, you can remove it.  Axe automatically adds 
the return for you so there is no reason to have 2 returns in a row.  Saves 1 byte.

Never make an if or while statement end with "not equal to zero".  Instead, you can omit  
that part since not being zero is implied.  Likewise, don't end an if,  repeat, or while  
statement with "equals zero".

Unoptimized
:If A≠0
:If A=0
:While A=0

Optimized
:If A
:!If A
:Repeat A

If  speed is super super important, you can chain divisions by 2 to quickly divide by 
higher powers of two.  It could increase the file size, but it will definitely be much faster. 
For instance, a faster way to do:  A/8  is:  A/2/2/2.

Animated  sprites  should  be  done  with  pointers.   Don't  use  a  huge  lists  of  if-then 
statements to decide which sprite to display.  Just define all the sprites in order one after  
another and reference them by offset from the first sprite.

Unoptimized
:If A=0:Pt-On(X,Y,Pic0):End
:If A=1:Pt-On(X,Y,Pic1):End
:If A=2:Pt-On(X,Y,Pic2):End

Optimized
:Pt-On(X,Y,A*8+Pic0)

There is a procedure called “Tail Call Optimization” which can save bytes and speed 
when you have a subroutine call another subroutine right before it returns.  Instead, you 
can just goto to that subroutine and steal its “Return” instead of coming back and using 
the original one.

Unoptimized
:sub(A)
:Return

Optimized
:Goto A

Other Info & Resources
The best  way I  think to  learn more about  the commands at  the moment is to look 
through the commands list file and play around with modifying the example programs. 
Just experiment with different things.

For specific help about specific questions, please I would prefer if you did not email me,  
instead I  have a forum dedicated to the parser at  Omnimaga.org.   The Axe Parser 
Forum is currently located at http://axe.omnimaga.org.  Its also a great resource to look 
through the topics there because you might find some very useful  information that I 
probably forgot about or skipped in this tutorial.  I usually have a new update there on a 
weekly basis.

If you want to contact me through email for some other reason, my email is:
compfreakkev@yahoo.com

I'm especially interested in bugs if you find any.  Although, you can also report that on 
the Axe Forums.  Make sure you have the latest version of Axe Parser before reporting.

Page 22

http://axe.omnimaga.org/
mailto:compfreakkev@yahoo.com


                             Axe Parser  1.1.2      

Credits
First  off,  thanks to the Omnimaga community for their  extraordinary help with alpha 
testing, giving me very helpful suggestions and feedback, and keeping me motivated to 
work on this project.  I couldn't have do it without you guys.

Special Thanks:

Brendan Fletcher  (calc84maniac) For his super savvy assembly knowledge and help 
with creating efficient routines for the parser.

Alex Marcolina (BuilderBoy) For the logo graphics and really showing Axe's full potential  
with awesome programs and screen shots.

Drew DeVault (SirCmpwn) For helping others with detailed tutorials online.

Brandon Wilson (BrandonW) Helped hugely with many of the awesome features that I  
would never be able to do without his l33t hacking skills.

Christopher Mitchell (Kerm Martian) Thanks for the help with the error scrolling!

Zachary Wassall (Runer112)  Helped tremendously with  finding new optimizations to 
help reduce the size of generated programs.

Etienne Gauvin Redesigned the entire HTML layout for the commands list to make it 
more aesthetically pleasing and printer friendly.

And of course giving credits to xkcd for the comics used in this tutorial.  Its my favorite 
web comic.  You'll probably like it too if you're nerdy (which clearly you are if you're  
reading a tutorial about how to program on a calculator).

Page 23

http://xkcd.com/

